Стоит очень древняя, очень сильно переписанная конфигурация УТ 10.2
В свое время я эту конфигурацию перевел с платформы 8.1 на платформу 8.2
Конфигурация при переводе осталась в режиме совместимости "Версия 8.1"
Со временем я переделал все на перевод в режим совместимости "Версия Версия 8.2.13".
На новогодних перевел конфигурацию на новый режим совместимости.
В результате в одном справочнике перестало корректно работать RLS

В ограничении доступа написан такой код:

```1c
    #Если &ИспользоватьОграниченияПравДоступаККонтрагентамНаУровнеЗапи­сей #Тогда
    ДокументацияПоСистеме ГДЕ ВЫБОР
    КОГДА &ЕстьПолныеПрава
    ИЛИ ДокументацияПоСистеме.ЭтоГруппа
    ТОГДА ИСТИНА
    ИНАЧЕ ВЫБОР
    КОГДА НЕ ДокументацияПоСистеме.ЕстьОграничениеПоДоступу
    ТОГДА ИСТИНА
    КОГДА &ОтделПользователя = ДокументацияПоСистеме.ГруппыДоступа.ГруппаПользователей
    ТОГДА ИСТИНА
    КОГДА &ТекущийПользователь = ДокументацияПоСистеме.ГруппыДоступа.ГруппаПользователей
    ТОГДА ИСТИНА
    ИНАЧЕ ЛОЖЬ
    КОНЕЦ
    КОНЕЦ
    #КонецЕсли
```

До перевода все работало без проблем, а сейчас пользователи ничего не видят, даже группы.
В чем может быть причина?
---
Нашел, в чем проблема.
В справочнике есть таблица "ГруппыДоступа" и в эту таблицу указывают тех, кто имеет доступ к конкретному элементу.
В 8.1 по своему отрабатывала эту ситуацию и поэтому выводилось все, что нужно.
В 8.2 это исправили и в запрос попадали только те элементы, в таблице которых были нужные пользователи.
Пришлось на скорую руку переписывать.
Получилось вот такое:

```1c
    #Если &ИспользоватьОграниченияПравДоступаККонтрагентамНаУровнеЗаписей #Тогда
    ДокументацияПоСистеме ГДЕ ВЫБОР
    КОГДА &ЕстьПолныеПрава
    ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.ЭтоГруппа
    ТОГДА ИСТИНА
    ИНАЧЕ ВЫБОР
    КОГДА НЕ ДокументацияПоСистеме.ЕстьОграничениеПоДоступу
    ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.Ссылка В
    (ВЫБРАТЬ
    ДокументацияПоСистемеГруппыДоступа.Ссылка
    ИЗ
    Справочник.ДокументацияПоСистеме.ГруппыДоступа КАК ДокументацияПоСистемеГруппыДоступа
    ГДЕ
    (ДокументацияПоСистемеГруппыДоступа.ГруппаПользователей = &ОтделПользователя
    ИЛИ ДокументацияПоСистемеГруппыДоступа.ГруппаПользователей = &ТекущийПользователь))
    ТОГДА ИСТИНА
    ИНАЧЕ ЛОЖЬ
    КОНЕЦ
    КОНЕЦ
    #КонецЕсли
```

Это заработало.
---
из вероятного
параметры криво устанавливаются
если режим иерархии поставить "Список" и ничего не видно, то см. раздел невероятное

невероятное
СУБД кривая
Платформа кривая
Сам код кривой
---
В коде прописано:
```1c
    КОГДА &ЕстьПолныеПрава
    ИЛИ ДокументацияПоСистеме.ЭтоГруппа
    ТОГДА ИСТИНА
```

Т.е. для групп всегда должно чтение произойти, а его не происходит.

Открытие списка справочника происходит всегда в иерархическом просмотре.

Опять же раньше работало а сейчас перестало.
---
попробуйте убрать иерархический просмотр и потом по результатам думать дальше, хотя там думать только о главном - о корне.
---
уберите ересь с ИЛИ. В этом случае оно неуместно

```1c
    ГДЕ ВЫБОР
    КОГДА &ЕстьПолныеПрава ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.ЭтоГруппа ТОГДА ИСТИНА
    ИНАЧЕ ВЫБОР....
```
---
Пробовал, не работает.
Пустой справочник открывается.
Пока не испробовал все, что сам мог придумать сюда не писал.
---
создайте в конфигурации новые формы списков(временно)
 - в виде простого списка без иерархии
 - в виде дерева

с какой-то версии платформы в палитре свойств списка есть свойство "Отображать корень" с ним по играйте.

если результат нет - переходим к разделу неверятное и ждем потного гарри.
---
Формы неуправляемые. Тут нет свойства отображать корень.

В справочнике контрагентов стоит точно такое же условие и все работает. Система видит все группы :(
---
Загрузил под полными правами, запустил консоль запросов.
создал вот такой запрос:

```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    ДокументацияПоСистеме.Ссылка,
    ДокументацияПоСистеме.Код,
    ДокументацияПоСистеме.Наименование,
    ДокументацияПоСистеме.ЭтоГруппа
    ИЗ
    Справочник.ДокументацияПоСистеме КАК ДокументацияПоСистеме
    ГДЕ
    ВЫБОР
    КОГДА &ЕстьПолныеПрава
    ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.ЭтоГруппа
    ТОГДА ИСТИНА
    //ИНАЧЕ ВЫБОР
    // КОГДА НЕ ДокументацияПоСистеме.ЕстьОграничениеПоДоступу
    // ТОГДА ИСТИНА
    // КОГДА &ОтделПользователя = ДокументацияПоСистеме.ГруппыДоступа.ГруппаПользователей
    // ТОГДА ИСТИНА
    // КОГДА &ТекущийПользователь = ДокументацияПоСистеме.ГруппыДоступа.ГруппаПользователей
    // ТОГДА ИСТИНА
    // ИНАЧЕ ЛОЖЬ
    // КОНЕЦ
    КОНЕЦ
```

В параметрах указал ЕстьПолныеПрава = Ложь;

В результате в отчет вывелись все группы.

Создал вот такой запрос:

```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    ДокументацияПоСистеме.Ссылка,
    ДокументацияПоСистеме.Код,
    ДокументацияПоСистеме.Наименование,
    ДокументацияПоСистеме.ЭтоГруппа
    ИЗ
    Справочник.ДокументацияПоСистеме КАК ДокументацияПоСистеме
    ГДЕ
    ВЫБОР
    КОГДА &ЕстьПолныеПрава
    ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.ЭтоГруппа
    ТОГДА ИСТИНА
    ИНАЧЕ ВЫБОР
    КОГДА НЕ ДокументацияПоСистеме.ЕстьОграничениеПоДоступу
    ТОГДА ИСТИНА
    КОГДА &ОтделПользователя = ДокументацияПоСистеме.ГруппыДоступа.ГруппаПользователей
    ТОГДА ИСТИНА
    КОГДА &ТекущийПользователь = ДокументацияПоСистеме.ГруппыДоступа.ГруппаПользователей
    ТОГДА ИСТИНА
    ИНАЧЕ ЛОЖЬ
    КОНЕЦ
    КОНЕЦ
```

В параметрах указал:
ЕстьПолныеПрава - ложь
ОтделПользователя - указал отдел, который точно есть в одном из элементов
ТекущийПользователь - указал пользователя, который тоже присутствует.
Выполнил запрос. В результате не вывелось ни одной группы, а вывелись только элементы, для которых указан отдел или пользователь.

Куда пропали группы???
---
Нашел, в чем проблема.
В справочнике есть таблица "ГруппыДоступа" и в эту таблицу указывают тех, кто имеет доступ к конкретному элементу.
В 8.1 по своему отрабатывала эту ситуацию и поэтому выводилось все, что нужно.
В 8.2 это исправили и в запрос попадали только те элементы, в таблице которых были нужные пользователи.
Пришлось на скорую руку переписывать.
Получилось вот такое:

```1c
    #Если &ИспользоватьОграниченияПравДоступаККонтрагентамНаУровнеЗаписей #Тогда
    ДокументацияПоСистеме ГДЕ ВЫБОР
    КОГДА &ЕстьПолныеПрава
    ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.ЭтоГруппа
    ТОГДА ИСТИНА
    ИНАЧЕ ВЫБОР
    КОГДА НЕ ДокументацияПоСистеме.ЕстьОграничениеПоДоступу
    ТОГДА ИСТИНА
    КОГДА ДокументацияПоСистеме.Ссылка В
    (ВЫБРАТЬ
    ДокументацияПоСистемеГруппыДоступа.Ссылка
    ИЗ
    Справочник.ДокументацияПоСистеме.ГруппыДоступа КАК ДокументацияПоСистемеГруппыДоступа
    ГДЕ
    (ДокументацияПоСистемеГруппыДоступа.ГруппаПользователей = &ОтделПользователя
    ИЛИ ДокументацияПоСистемеГруппыДоступа.ГруппаПользователей = &ТекущийПользователь))
    ТОГДА ИСТИНА
    ИНАЧЕ ЛОЖЬ
    КОНЕЦ
    КОНЕЦ
    #КонецЕсли
```

Это заработало.